

21 *Software and Tools*

After this chapter you will be able to:

1. state which of FV, PV, PMT, RATE, and NPER solves for each unknown, and read Excel's sign convention correctly
2. build a loan amortization schedule from PMT and three filled-down formulas, and check that the final balance is zero
3. distinguish NPV, whose range starts at period 1, from IRR, whose range includes year 0
4. compute depreciation schedules with SLN, DDB, and VDB, and build a MACRS schedule from a typed percentage table
5. choose a one-variable Data Table for one-way sensitivity and Goal Seek for a one-unknown break-even problem
6. read the economic service life N^* from an $AEC(N)$ table built with two PMT calls per row

Every problem in this book so far was solved with factor tables, on purpose. A factor solution keeps the reasoning visible: the cash flow diagram shows *what* moves, the factor notation shows *how* it moves, and the table lookup is a step you can audit. That is the right way to learn the subject, and it remains the required method on graded work. Professional practice, however, runs on spreadsheets. An engineer who evaluates a flotation-circuit upgrade builds the cash flow table in Excel, not on a factor-table page. This chapter maps every method of Chapters 2–16 to its spreadsheet function, using problems you have already solved by hand, so that every result can be checked against an answer you trust. The closing sections then carry the same program forward to methods that arrive later in the book: simulation (chapter 14), decision trees (chapter 13), levelized cost with a carbon price (chapter 17), and cost estimation (chapter 9).

The functions are shown in Microsoft Excel syntax; LibreOffice Calc and Google Sheets accept the same names and arguments.

Seven companion workbooks are distributed with the book, in the `spreadsheets/` folder:

- `01_time_value_of_money.xlsx` — the five time-value functions and a full loan amortization schedule;
- `02_project_evaluation.xlsx` — NPV and IRR, a present-worth profile chart, and incremental analysis;
- `03_depreciation_taxes.xlsx` — SLN, DDB, and VDB schedules, a MACRS schedule, and an after-tax cash-flow table;
- `04_uncertainty_replacement.xlsx` — one-way sensitivity with a Data Table, break-even with Goal Seek, a scenario table, and an economic-service-life AEC table;
- `05_simulation_and_trees.xlsx` — a 1,000-row Monte Carlo simulation of the crushing unit, with percentiles, a loss probability, and a histogram, and a decision-tree rollback with an EVPI cell;
- `06_levelized_cost.xlsx` — LCOE calculators for diesel and solar-plus-storage, the same comparison under an internal carbon price, and a two-pump life-cycle-cost table;
- `07_estimation.xlsx` — capacity scaling by the six-tenths rule, escalation by index ratio, and a contingency lookup from an AACE-style class table.

Each section below names the workbook and the sheet it uses, so you can open the finished version next to the text, or rebuild it from a blank sheet, which is the better exercise.

One expectation, stated plainly. The exam still requires the hand method: diagram, factor notation, table lookup, boxed answer. The spreadsheet is for checking your hand work and for the real projects that come after this course. Learn both, and keep them in that order.

Signs, ranges, and off-by-one years.

Q1. You type `=PV(0.10, 5, 4000)` and Excel returns `-15,163.15`. Is the minus sign an error?

- Q2.** A project's cash flows sit in cells A1:A6, with the year-0 investment in A1. What is wrong with `=NPV(0.10, A1:A6)`?
- Q3.** True or false: the IRR function needs the year-0 cell inside its range, and it assumes the cash flows are equally spaced in time.

Answers.

Q1. No. Excel's time-value functions use a sign convention: money you receive is positive, money you pay is negative. The formula asks what present amount is equivalent to *receiving* 4,000 SAR per year for five years at 10%. The answer is money you would have to *pay* now, so it comes back negative. The magnitude matches the hand result, $4,000 (P/A, 10\%, 5) = 4,000 (3.7908) = 15,163$ SAR.

Q2. The NPV function discounts the *first* cell of its range by one period. It assumes the range starts at year 1, not year 0. The formula above discounts the investment as if it happened a year from now, which overstates the present worth of a cost (and understates it for the whole project). The correct form is `=A1 + NPV(0.10, A2:A6)`: the year-0 flow is added outside the function, at face value.

Q3. Both are true. `=IRR(A1:A6)` treats the first cell as time zero and each following cell as one period later, so the investment must be *inside* the range — the opposite of the NPV rule — and a skipped year must be entered as an explicit zero, never as an empty cell. Unevenly dated flows need a different function (XIRR), which is outside the scope of this course.

21.1 The Five Time-Value Functions

Reference. Companion workbook 01_time_value_of_money.xlsx, the five-functions sheet.

Excel has one function for each unknown in the equivalence equation of Chapter 2. All five share one family of arguments:

rate	the interest rate per period, as a decimal
nper	the number of periods N
pmt	the equal payment A per period (o if none)
pv	the present amount P (o or omitted if none)
fv	the future amount F (o or omitted if none)
type	o for end-of-period flows (the default; always o in this book)

Each function takes the knowns in a fixed order and solves for the one quantity it is named after:

Sign convention. In Excel’s time-value functions, cash you receive is positive and cash you pay out is negative. Every argument and every result obeys it.

In practice. A formula that computes is not a model that is right. The habit that transfers from engineering is the order-of-magnitude check: a loan payment should sit near the principal divided by the years, and a present worth should sit below the undiscounted total. A result that fails such a check is wrong before any cell is audited.

=FV(rate, nper, pmt, [pv], [type])	finds F
=PV(rate, nper, pmt, [fv], [type])	finds P
=PMT(rate, nper, pv, [fv], [type])	finds A
=RATE(nper, pmt, pv, [fv], [type])	finds i
=NPER(rate, pmt, pv, [fv], [type])	finds N

Because one function accepts P , A , and F together, a single call can replace a two-factor hand solution. The price of this power is a strict sign discipline.

Insight. The sign convention is the single most common spreadsheet error in this course, so walk through the trap once. A student wants the future worth of a 100,000 SAR deposit at 7% for 8 years and types =FV(0.07, 8, 0, 100000). Excel returns −171,818.62. The student, seeing a negative where a positive was expected, deletes the formula and computes $100,000 \times 1.07^8$ by hand instead. But Excel was not wrong; the *question* was entered wrong. A positive pv means money received now, a loan, and the negative result is the repayment owed at year 8. A deposit is money paid out now, so it must be entered as −100,000: =FV(0.07, 8, 0, -100000) returns +171,818.62, the withdrawal available at year 8. Do not fight the signs by sprinkling minus signs until the answer looks right. Decide the direction of each flow first, exactly as you do when drawing a cash flow diagram: the convention is the CFD, with down arrows negative and up arrows positive.

The five short examples below reproduce hand-solved problems from Chapter 2, one function at a time. The grid tables show the spreadsheet exactly as typed: column letters across, row numbers down.

Example 21.1 — FV: the robotics research fund again

In Example 2.9, a robotics research fund of 100,000 SAR invested at 7% for 8 years grew to 100,000 $(F/P, 7\%, 8) = 100,000 (1.7182) = 171,820$ SAR. Reproduce the result with FV.

Solution.

Enter the inputs in their own cells and point the formula at them:

	A	B
1	rate	0.07
2	years	8
3	deposit now	-100000
4	future worth	=FV(B1, B2, 0, B3)

Table 21.1. The FV grid for the research fund. The deposit is entered negative because it is money paid out.

Cell B4 displays 171,818.62. The `pmt` argument is 0 because there is no annual series, and the deposit enters as $-100,000$: money out.

$$F = 171,818.62 \text{ SAR}$$

The hand answer was 171,820 SAR. The 1.38 SAR gap is not an error; it is the table’s rounding of (F/P) to four decimals, a point taken up at the end of this section.

Example 21.2 — PV: discounting a future amount again

In Example 2.8, the present worth of 150,000 SAR due in seven years at 6% was 150,000 $(P/F, 6\%, 7) = 150,000 (0.6651) = 99,765$ SAR. Reproduce the result with PV.

Solution.

	A	B
1	rate	0.06
2	years	7
3	amount received at year 7	150000
4	present worth	=PV(B1, B2, 0, B3)

Table 21.2. The PV grid. The future amount is positive (money in), so the present equivalent comes back negative (money out).

Cell B4 displays $-99,758.57$: to be indifferent to receiving 150,000 SAR in seven years, you would pay

$$P = 99,758.57 \text{ SAR}$$

today. Read the sign as information, not as a nuisance: it states which direction the equivalent flow points. The hand answer, 99,765 SAR, is 6.43 SAR higher because the table rounds the exact (P/F) factor, 0.665057 . . . , up to 0.6651.

Example 21.3 — PMT: a loan payment and a sinking fund

Reproduce two hand results with one function. (a) In Example 2.16, a 50,000 SAR conveyor loan at 8% was repaid in five equal installments of 50,000 ($A/P, 8\%, 5$) = 12,523 SAR. (b) In Example 2.11, a 100,000 SAR overhaul in 8 years at 7% was funded by deposits of 100,000 ($A/F, 7\%, 8$) = 9,747 SAR per year.

Solution.

PMT covers both, because its third and fourth arguments accept a present amount, a future amount, or both:

	A	B	C
1	purpose	formula	displays
2	loan installment (A/P)	=PMT(0.08, 5, 50000)	-12,522.82
3	sinking-fund deposit (A/F)	=PMT(0.07, 8, 0, 100000)	-9,746.78

Table 21.3. One function, two factors. Row 2 uses the pv argument; row 3 skips pv with a 0 and uses fv.

In row 2, the loan of 50,000 SAR is money received (positive), so the payments are money out (negative):

$$A = 12,522.82 \text{ SAR per year}$$

against 12,523 SAR by hand. In row 3, the target 100,000 SAR is entered as a positive future receipt, and the required deposits come back negative:

$$A = 9,746.78 \text{ SAR per year}$$

against 9,747 SAR by hand. The factor pair (A/P) and (A/F) from Chapter 2 is one spreadsheet function with two different arguments filled in.

Example 21.4 — RATE: the equivalence rate found by iteration

In exercise Exercise 2.12, a foundation compared 200,000 SAR now with 28,000 SAR per year for 12 years. The hand solution set $(P/A, i, 12) = 7.1429$ and interpolated between the 9% and 10% table pages to get $i \approx 9.05\%$. Reproduce the rate with RATE.

Solution.

	A	B
1	years	12
2	annual amount (received)	28000
3	lump sum (given up)	-200000
4	equivalence rate	=RATE(B1, B2, B3)

Table 21.4. The RATE grid. Taking the payment stream means giving up the lump sum, so the 200,000 SAR enters negative and the 28,000 SAR positive.

Cell B4 displays 9.0496%:

$$i = 9.05\%$$

in agreement with the hand interpolation to the second decimal. Two notes. First, RATE finds the rate by iteration, the same bracket-and-refine idea as the trial-and-error method of Chapter 7, only automated; for unusual cash flows it accepts a starting guess as an extra argument. Second, the signs must disagree: if both amounts are entered positive, there is no rate at which money only flows in, and the function returns an error.

Example 21.5 — NPER: how long to double

Section 2.9 showed that money at 12% doubles in $N = \ln 2 / \ln 1.12 = 6.12$ years, and the Rule of 72 estimates $72/12 = 6$ years. Reproduce the exact figure with NPER.

Solution.

	A	B
1	rate	0.12
2	pay 1 now	-1
3	receive 2 later	2
4	periods to double	=NPER(B1, 0, B2, B3)

Table 21.5. The NPER grid. Any pair of amounts in ratio 1:2 works; the function only sees the ratio.

Cell B4 displays 6.1163:

$$N = 6.12 \text{ years}$$

matching the logarithm solution. Note that NPER returns a fractional period. For a decision such as “in which year does the fund first reach the target,” round *up* to the next whole period, because the balance passes the target during that period, and end-of-period flows only settle at its end.

Insight. Why Excel and the tables disagree in the last digits. The factor tables print each factor rounded to four or five significant decimals; Excel carries about fifteen. Multiplying a rounded factor by a large amount scales the rounding up: a 5×10^{-5} error in (P/F) times 150,000 SAR is about 7 SAR. So table answers and spreadsheet answers for the same problem routinely differ by a few SAR in the last digits, exactly as the preface warned. Neither is wrong. On graded work, the table value is the expected answer; in the workbooks, the exact value is. When you check hand work with a spreadsheet, accept differences of this size, and investigate only differences that are far too large to be rounding, because those usually mean a sign error or a misplaced year.

Reference. Companion workbook 01_time_value_of_money.xlsx, the loan-amortization sheet.

Amortization schedule. A year-by-year table of a loan: payment, interest on the opening balance, principal repaid, and closing balance. The final balance must be exactly zero.

21.2 Loan Schedules

Chapter 2 showed that every equal loan payment splits into interest on the outstanding balance plus a principal repayment, and that the split shifts from interest to principal as the balance falls. A spreadsheet makes that statement into a table you can read, and it is the standard deliverable when an engineer presents financed equipment. This section builds the full schedule for the conveyor loan

of Example 2.16: 50,000 SAR at 8% over 5 years.

Example 21.6 — The conveyor loan, year by year

Build the full amortization schedule for the 50,000 SAR, 8%, 5-year conveyor loan, and verify that the final balance is zero.

Solution.

Step 1: the inputs and the payment.

Give every input its own labeled cell, then compute the payment once:

	A	B
1	loan amount	50000
2	rate	0.08
3	years	5
4	payment	=PMT(B2, B3, -B1)

Table 21.6. The input block of the loan sheet. Entering the loan as -B1 inside PMT flips the convention so the payment displays positive, which is more natural inside a cost table.

Cell B4 displays 12,522.82, the exact value of the hand answer 12,523 SAR.

Step 2: the schedule.

Lay out five columns: year, payment, interest, principal, and balance. Seed the year-0 row with the opening balance, then write the year-1 row entirely from formulas:

cell	formula
E7 (year-0 balance)	=B1
B8 (payment)	=B\$4
C8 (interest)	=B\$2*E7
D8 (principal)	=B8 - C8
E8 (balance)	=E7 - D8

The interest is the rate times the *previous* balance; the principal is whatever part of the payment the interest does not consume; the new balance is the old balance less the principal. Now select B8:E8 and fill down four rows. The relative references E7, B8, C8 shift down a row each time, which is exactly the recursion, and the anchored B\$4 and B\$2 keep pointing at the payment and the rate. The result:

	A	B	C	D	E
7	0				50,000.00
8	1	12,522.82	4,000.00	8,522.82	41,477.18
9	2	12,522.82	3,318.17	9,204.65	32,272.53
10	3	12,522.82	2,581.80	9,941.02	22,331.51
11	4	12,522.82	1,786.52	10,736.30	11,595.21
12	5	12,522.82	927.62	11,595.21	0.00

Table 21.7. The amortization schedule for the conveyor loan (columns: year, payment, interest, principal, balance). Interest falls and principal grows, while the payment stays constant.

Step 3: the checks.

Three checks cost nothing and catch almost every construction error. The final balance E12 is exactly 0.00; if it is not, a reference slipped during the fill. The principal column sums to the loan: =SUM(D8:D12) returns 50,000.00. The interest column sums to the total interest: =SUM(C8:C12) returns 12,614.11 SAR, against the hand estimate $5 \times 12,523 - 50,000 = 12,615$ SAR in Example 2.16, the usual last-digit rounding.

Excel also answers schedule questions without building the table. =IPMT(rate, per, nper, pv) returns the interest part of one payment, =PPMT(...) the principal part, and =CUMIPMT(rate, nper, pv, start, end, 0) the interest accumulated between two payments:

question	formula	displays
interest in year 1	=IPMT(0.08, 1, 5, -50000)	4,000.00
principal in year 1	=PPMT(0.08, 1, 5, -50000)	8,522.82
interest in year 3	=IPMT(0.08, 3, 5, -50000)	2,581.80
all interest, years 1–5	=CUMIPMT(0.08, 5, 50000, 1, 5, 0)	-12,614.11

Each value matches its column in Table 21.7. Note the quirk in the last row: CUMIPMT insists on a positive pv and returns the interest as a negative amount (money out); its final argument is the type flag, 0 for end-of-period.

Reference. Companion workbook 02_project_evaluation.xlsx: the NPV–IRR sheet, the PW-profile sheet, and the incremental-analysis sheet.

21.3 Project Evaluation: NPV and IRR

The five functions of Section 21.1 handle regular patterns: one *P*, one *F*, one uniform *A*. Project cash flows are rarely that tidy, so the workhorse pair for

Chapters 5 and 7 takes a whole *range* of cells instead: NPV for present worth and IRR for the rate of return. Both read the range in order and assume one period between neighboring cells.

Example 21.7 — NPV: the haul-truck retrofit again

In Example 5.3, the autonomy retrofit (−55,000 SAR at year 0, +16,000 SAR in years 1–4, +24,000 SAR in year 5) had $PW(12\%) = +7,216$ SAR by hand. Reproduce the result, and the IRR, in one column.

Solution.




Figure 21.1. The retrofit flows mapped to cells: the year-0 arrow lives in B3, outside the NPV range; years 1–5 live in B4:B8, inside it.

	A	B
1	MARR	0.12
2	year	cash flow
3	0	-55000
4	1	16000
5	2	16000
6	3	16000
7	4	16000
8	5	24000
9	PW	=B3+NPV(B1, B4:B8)
10	IRR	=IRR(B3:B8)

Table 21.8. The evaluation column for the retrofit. Note the asymmetry: NPV’s range starts at year 1 and the year-0 cell is added outside; IRR’s range starts at year 0.

Cell B9 displays 7,215.83:

$$\boxed{\text{PW}(12\%) = +7,215.83 \text{ SAR}} \Rightarrow \text{accept,}$$


against +7,216 SAR by hand. Cell B10 displays 16.88%, above the 12% MARR, so the IRR rule of Chapter 7 gives the same verdict, as it must for a simple investment.

Insight. The name NPV is a trap and the error it causes is the classic one of this chapter. Despite its name, NPV does *not* compute the net present worth of the range you give it as if it started today. It discounts the first cell one full period, the second cell two periods, and so on: the function assumes its range starts at the *end of period 1*. Typing =NPV(B1, B3:B8) for the retrofit returns 6,442.71 SAR, which is the correct answer divided by 1.12: every flow, including the investment, slid one year into the future. The correct pattern is always

$$=A0 + \text{NPV}(\text{rate}, A1:AN),$$

with the year-0 flow added at face value outside the function. IRR has the opposite convention: its range must *include* the year-0 cell. Check both every time; the two functions sitting one row apart with different ranges is normal and correct.

Figure 21.2 shows the two ranges against the column of Table 21.8; check which cell each range starts at, because that single row is the whole difference between the two answers.



IRR runs the other way: `=IRR(B3:B8)` must *include* the year-o cell.

Figure 21.2. The year-zero trap in the spreadsheet NPV function. NPV discounts its first cell one full period, so a range that starts at the year-o cell pushes every flow one year late and returns the right answer divided by $1+i$ — 6,442.71 instead of 7,215.83 SAR. The year-o cell belongs outside the function, added at face value.

Example 21.8 — The PW profile as a rate column

In Example 7.3, the conveyor drive (−75,000 SAR, then +24,400 SAR for 5 years) was bracketed by hand: $PW(18\%) = +1,304$, $PW(20\%) = -2,029$, interpolating to $i^* \approx 18.78\%$. Build the full present-worth profile and read the exact root.

Solution.

Put the flows in B3:B8 (year o in B3) as before. Then list candidate rates in a column and evaluate the same PW formula at each one: in D3 down to D9 type the rates, and in E3 type

`=B$3 + NPV(D3, $B$4:$B$8)`

and fill down. The dollar signs anchor the cash flow range while the rate reference D3 walks down the column — one formula, seven rates:

	D	E
3	0%	47,000.00
4	5%	30,639.23
5	10%	17,495.20
6	15%	6,792.58
7	20%	−2,029.06
8	25%	−9,381.57
9	30%	−15,572.10

Table 21.9. The PW profile of the conveyor drive as a rate column. The sign changes between 15% and 20%, the same bracket the hand method found.

Select the two columns and insert a scatter chart to get the profile curve of Chapter 7 exactly; the companion workbook includes the chart. Finally, `=IRR(B3:B8)` returns

$i^* = 18.76\%$

the exact root. The hand interpolation gave 18.78%, high by two hundredths because the straight chord sits under the convex profile, precisely as the insight in Chapter 7 predicted. Note also the small table-rounding shifts in the profile itself: the hand ledger had $PW(15\%) = +6,793.68$ from four-decimal factors; the exact value is +6,792.58.

Example 21.9 — Incremental IRR as the IRR of a difference column

In Example 7.9, project A (−2,000 now, +3,600 in one year) earns 80% and project B (−8,000, +11,200) earns 40%, yet B is the right choice at a MARR of 10%. Reproduce the whole argument, including the incremental rate of Example 7.10, in one grid.

Solution.

	A	B	C	D
1	year	A	B	B−A
2	0	−2,000	−8,000	=C2-B2
3	1	3,600	11,200	=C3-B3
4	IRR	=IRR(B2:B3)	=IRR(C2:C3)	=IRR(D2:D3)
5	PW at 10%	=B2+NPV(0.1,B3)	=C2+NPV(0.1,C3)	

Table 21.10. The ranking paradox in five rows. Column D is built by subtraction, and its IRR is the incremental rate.

Row 4 displays 80%, 40%, and 26.67%; row 5 displays 1,272.73 and 2,181.82 (the hand values were 1,273 and 2,182 SAR from the rounded factor 0.9091). The difference column D holds −6,000 and +7,600, and

$$i_{B-A}^* = 26.67\% > 10\% = \text{MARR} \Rightarrow \text{choose B,}$$

agreeing with the present-worth ranking, exactly as Chapter 7 requires. The spreadsheet lesson is the method: build the increment as a literal column of =C-B formulas and apply IRR to it. Never subtract the two IRRs; the incremental rate is the rate of a difference, not a difference of rates. Before trusting the incremental rate, scan the difference column for its sign pattern, one change here, so the rate is trustworthy — the spreadsheet computes a number, but the classification duty from Chapter 7 stays with you.

21.4 Depreciation and Taxes

Excel ships with the book depreciation methods of Chapter 10 as ready functions, all sharing the argument head (cost, salvage, life, ...):

=SLN(cost, salvage, life)	the constant straight-line charge
=DDB(cost, salvage, life, period)	one year's declining-balance charge
=VDB(cost, salvage, life, start, end)	DDB with the switch to straight line

Reference. Companion workbook 03_depreciation_taxes.xlsx: the book-methods sheet, the MACRS sheet, and the ATCF sheet.

Example 21.10 — SLN and DDB: the flotation cell again

Rebuild the straight-line schedule of Example 10.2 and the DDB-with-salvage-floor schedule of Example 10.3 for the flotation cell: $I = 50,000$ SAR, $S =$

5,000 SAR, $N = 5$ years.

Solution.

Put the three inputs in B1:B3 (cost, salvage, life). In the schedule, year numbers run down column A from A6; the two charge columns are

B6: =SLN(\$B\$1, \$B\$2, \$B\$3) C6: =DDB(\$B\$1, \$B\$2, \$B\$3, A6),

filled down five rows. Every reference to the inputs is anchored with \$, so the fill-down changes only the year argument A6:

	A	B	C
5	year	SLN	DDB
6	1	9,000.00	20,000.00
7	2	9,000.00	12,000.00
8	3	9,000.00	7,200.00
9	4	9,000.00	4,320.00
10	5	9,000.00	1,480.00

Table 21.11. The flotation cell under SLN and DDB. Both columns sum to the depreciable amount of 45,000 SAR.

Both columns reproduce the hand schedules exactly, including the delicate year: DDB returns 1,480.00 in year 5, not the uncapped 2,592 SAR, because the function applies the same salvage floor as the book’s rule — it never depreciates below the salvage argument. That built-in cap is why the salvage must be typed correctly: with a salvage of 0, DDB would happily return 2,592 and keep going.

For the opposite complication, the switch to straight line when the salvage is low, use VDB. For the instrument bench of Example 10.4 (10,000 SAR, no salvage, 5 years), the year- n charge is

=VDB(10000, 0, 5, A6-1, A6),

the depreciation *between* ages $n - 1$ and n . Filled down, it returns 4,000, 2,400, 1,440, 1,080, 1,080: the same schedule as the hand switching rule, with the switch appearing in year 4 and the total reaching the full 10,000 SAR.

Absolute reference. A cell address locked with dollar signs, as in \$B\$1, so it does not shift when the formula is filled down or across.

There is no built-in MACRS function. Since MACRS is just a published percentage table (Table 10.7), the spreadsheet method is the hand method: type the percentages once and multiply.

Example 21.11 — A MACRS schedule with absolute references

Rebuild the 5-year MACRS schedule of Example 10.7 for the bagging equipment: cost basis 25,000 SAR.

Solution.

	A	B	C	D
1	basis	25000		
3	year	p_n	D_n	BV_n
4	1	20.00%	5,000.00	20,000.00
5	2	32.00%	8,000.00	12,000.00
6	3	19.20%	4,800.00	7,200.00
7	4	11.52%	2,880.00	4,320.00
8	5	11.52%	2,880.00	1,440.00
9	6	5.76%	1,440.00	0.00

Table 21.12. The MACRS sheet: the tax authority's percentages typed in column B, one multiplication in column C, a running book value in column D.

Column C holds $=B\$1*B4$ filled down: the basis stays anchored while the percentage reference walks. Column D holds $=B\$1 - \text{SUM}(\$C\$4:C4)$ filled down; the half-anchored range $\$C\$4:C4$ stretches as it fills, so each row subtracts the accumulated depreciation to date. The checks are the ones from Chapter 10: $=\text{SUM}(C4:C9)$ returns the full basis of 25,000 SAR, and the final book value is exactly zero, because MACRS treats salvage as zero.

Example 21.12 — The bagging line's after-tax table

Rebuild the after-tax evaluation of Example 10.12: a bagging line with cost basis 60,000 SAR, straight-line over 4 years to $S = 12,000$ SAR, revenue 50,000 SAR per year, cash expenses 20,000 SAR per year, actual sale for 15,000 SAR at year 4, tax rate 25%, MARR 10%.

Solution.

Step 1: the operating table.

One row per year, one column per line of the income-statement pipeline of Chapter 10, with the tax rate in the anchored input cell B2:

	A	B	C	D	E	F	G	H
4	<i>n</i>	<i>R</i>	<i>E</i>	<i>D</i>	TI	tax	NI	ATCF
5	1	50,000	20,000	12,000	18,000	4,500	13,500	25,500
6	2	50,000	20,000	12,000	18,000	4,500	13,500	25,500
7	3	50,000	20,000	12,000	18,000	4,500	13,500	25,500
8	4	50,000	20,000	12,000	18,000	4,500	13,500	25,500

Table 21.13. The operating ATCF table for the bagging line. Columns E–H are formulas; only the estimates in B–D are typed.

The row-5 formulas, filled down: depreciation D5: =SLN(60000, 12000, 4); taxable income E5: =B5-C5-D5; tax F5: =\$B\$2*E5; net income G5: =E5-F5; and the cash flow H5: =G5+D5, net income with the non-cash depreciation added back. Every year shows the hand value, 25,500 SAR.

Step 2: the disposal and the verdict.

The sale at 3,000 SAR above book value nets

$$15,000 - 0.25 (3,000) = 14,250 \text{ SAR,}$$

entered as a formula so a tax-rate change propagates. Assemble the final cash flow column in B12:B16: -60,000; three years of 25,500; and 39,750 in year 4, the operating flow plus the net proceeds. Evaluate the column with the standard pattern:

$$=B12+NPV(0.10, B13:B16).$$

The cell displays

PW(10%) = +30,564.51 SAR

 ⇒ **accept,**

against +30,565.20 SAR by hand, a 0.69 SAR factor-rounding difference. The structure is the payoff here: because the tax rate, the salvage estimate, and the

revenue all live in input cells, this sheet is ready for every what-if question of Chapter 12 without further surgery.

21.5 Uncertainty: Data Tables and Goal Seek

Chapter 12 varied the inputs of the Wadi Phosphate mobile crushing unit one at a time, found break-even values, and priced scenarios. Each tool has a direct spreadsheet counterpart, and this is where the one-change-propagates discipline pays off. The base sheet, used by all three examples below, is:

	A	B
1	MARR	0.10
2	life (years)	5
3	annual revenue	42000
4	operating cost	26000
5	investment	55000
6	salvage	5000
7	annual net	=B3-B4
8	PW	=-B5-PV(B1, B2, B7, B6)

Table 21.14. The crushing-unit base sheet. Row 8 uses PV with both a payment and a future value; the leading minus signs convert the paid-out convention into a plain present worth.

Cell B8 displays 8,757.19, the base case of Example 12.1 (+8,757 SAR by hand). Every input has its own cell and the PW is one formula away from all of them.

Reference. Companion workbook 04_uncertainty_replacement.xlsx: the sensitivity sheet, the break-even sheet, and the scenario sheet.

Data Table. Excel's what-if tool that re-evaluates a formula cell for each value in a row or column of candidate inputs, in one operation.

Example 21.13 — One-way sensitivity with a one-variable Data Table

Reproduce the revenue row of the sensitivity table in Example 12.2: the PW of the crushing unit with annual revenue at -20% , -10% , base, $+10\%$, and $+20\%$ of the 42,000 SAR estimate.

Solution.

Step 1: the layout.

A one-variable Data Table wants the candidate inputs in a column and the formula one cell up and one cell right of the column's first entry. In D3:D7, type the five revenue levels 33,600, 37,800, 42,000, 46,200, 50,400. In E2, link

to the output: =B8.

Step 2: the menu path.

Select the block D2:E7. Then: Data tab → What-If Analysis → Data Table. Leave “Row input cell” empty; set “Column input cell” to B3, the revenue input. Click OK. Excel substitutes each value of column D into B3, recomputes the sheet, and records B8:

	D	E
2		=B8
3	33,600	−23,085.41
4	37,800	−7,164.11
5	42,000	8,757.19
6	46,200	24,678.50
7	50,400	40,599.80

Table 21.15. The one-variable Data Table for revenue. Column D holds the candidate revenues; column E fills with the recomputed PW.

The five entries match the revenue row of Table 12.2, −23,085 / −7,164 / 8,757 / 24,679 / 40,600, to the SAR. Repeat with a second table pointed at B4 to reproduce the operating-cost row (28,469.29 at −20% down to −10,954.90 at +20%), and so on for each input; chart the columns together and the sensitivity graph of Chapter 12 appears. The reading is unchanged from Example 12.2: the sign of the PW already flips between −10% and base revenue, the steepest and most dangerous line in the graph.

Goal Seek. Excel’s one-unknown solver: it adjusts one input cell until a chosen formula cell reaches a stated value.

Example 21.14 — Break-even revenue with Goal Seek

In Example 12.4, the crushing unit’s break-even annual net was solved by hand: $A^* = 13,690$ SAR per year, a 14.4% cushion, which means break-even revenue $R^* = 39,690$ SAR, only 5.5% below forecast. Reproduce the break-even with Goal Seek.

Solution.

Break-even asks: what revenue makes B8 exactly zero? That is one unknown and one target, which is Goal Seek’s entire job. The menu path: Data

tab → What-If Analysis → Goal Seek, then fill the three boxes of the dialog:

Set cell	B8	(the PW formula)
To value	0	
By changing cell	B3	(the revenue input)

Click OK, and Excel iterates B3 until B8 reaches zero. The sheet settles at

$$R^* = 39,689.87 \text{ SAR per year}$$

so the break-even annual net is $39,689.87 - 26,000 = 13,689.87$ SAR, against the hand value 13,690 SAR, with the same cushions: 5.5% on revenue, 14.4% on the net. Two habits keep Goal Seek honest. First, “Set cell” must contain a formula and “By changing cell” must contain a plain number; pointing it at the formula cell B7 fails. Second, Goal Seek overwrites the input, so read the answer, then undo, or work on a copy of the sheet.

Example 21.15 — The scenario table

Reproduce the three-scenario analysis of Example 12.5: pessimistic, most likely, and optimistic stories for the crushing unit.

Solution.

Scenarios change several inputs at once, so a Data Table (one input) and Goal Seek (one unknown) do not fit. The plain solution is one row per story, with the PW recomputed from that row’s own inputs:

	A	B	C	D	E
11	scenario	revenue	op. cost	salvage	PW
12	pessimistic	36,000	28,000	3,000	−22,810.94
13	most likely	42,000	26,000	5,000	8,757.19
14	optimistic	46,000	24,000	7,000	32,743.76

Table 21.16. The scenario table: one coherent story per row, one PW per story.

Cell E12 holds $= -\$B\$5 - PV(\$B\$1, \$B\$2, B12 - C12, D12)$, filled down: the investment, MARR, and life stay anchored to the base block, while the revenue, cost, and salvage come from the scenario’s own row. The three present worths

Reference. Companion workbook 04_uncertainty_replacement.xlsx, the economic-service-life sheet.

match Example 12.5: $-22,811$, $+8,757$, and $+32,744$ SAR by hand. Excel also offers a Scenario Manager under the same What-If Analysis menu, but the plain table is easier to audit, and auditability is worth more than the menu.

21.6 Replacement: The AEC Table

The economic-service-life table of Chapter 16 is the most formula-dense hand computation in the book: every row needs its own (A/P) , a capital recovery, and an annual equivalent of a growing operating-cost stream. It is also where the spreadsheet earns its keep, because the whole of Equation (16.1) collapses into two PMT calls per row.

Example 21.16 — The ore-transfer truck’s AEC(N) table

Rebuild the challenger truck’s economic-service-life table of Example 16.4: $I = 72,000$ SAR, $i = 10\%$, with the market values and operating costs estimated year by year.

Solution.

Step 1: inputs and layout.

The price and rate go in B1:B2; the year-by-year estimates S_N and OC_N go in columns B and C of the table:

	A	B	C	D	E	F
4	N	S_N	OC_N	CR	op. AE	$AEC(N)$
5	1	54,000	8,000	25,200.00	8,000.00	33,200.00
6	2	42,000	11,000	21,485.71	9,428.57	30,914.29
7	3	33,000	15,000	18,982.48	11,111.78	30,094.26
8	4	26,000	20,000	17,111.66	13,026.93	30,138.59

Table 21.17. The AEC(N) sheet for the ore-transfer truck. Columns D–F are three formulas filled down four rows.

Step 2: capital recovery by PMT.

In D5:

$$=PMT(\$B\$2, A5, -\$B\$1, B5),$$

filled down. Read the arguments against the ownership story: pay the price now ($p_v = -72,000$), hold for A_5 years, and recover the resale value at the end ($f_v = S_N$, money in). The function returns the equal annual amount that balances that account, which is exactly the capital recovery cost $(I - S_N)(A/P, i, N) + i S_N$. No separate (A/P) and (A/F) lookups per row.

Step 3: the operating annual equivalent.

In E5:

`=PMT($B$2, A5, -NPV($B$2, $C$5:C5)),`

filled down. The inner NPV discounts the operating costs of years 1 through N to year 0 — the half-anchored range $\$C\$5:C5$ stretches as it fills, taking in one more year per row — and the outer PMT spreads that present amount back over the same N years. This is the two-step $(\sum OC_n(P/F))(A/P)$ of Equation (16.1), verbatim.

Step 4: total and flag.

F5: `=D5+E5`, filled down. Then two closing formulas find the minimum and its location:

`=MIN(F5:F8) → 30,094.26,      =MATCH(MIN(F5:F8), F5:F8, 0) → 3.`

MATCH with final argument 0 returns the position of the exact minimum within the range; since the table starts at $N = 1$, position 3 means

$$N^* = 3 \text{ years, } AEC^* = 30,094.26 \text{ SAR per year}$$

the same verdict as Table 16.2, whose hand values (30,093.7 at $N = 3$, 30,138.2 at $N = 4$) sit within a fraction of a percent — four-decimal factors again. The practical gain shows up the day an estimate changes: when the year-3 market value is revised, one cell edit reprices the whole policy table, flag included.

Reference. Companion workbook 05_simulation_and_trees.xlsx, the MonteCarlo sheet.

21.7 Simulation in a Spreadsheet

Chapter 12 varied the crushing unit's inputs one at a time and priced three scenarios. chapter 14 replaces the handful of stories with thousands of random ones and reads the answer as a distribution. A spreadsheet needs only two ingredients for that: a source of randomness and a formula that shapes it into a

Volatile function. A function, such as `RAND()`, that returns a fresh value on every recalculation of the sheet. Pressing F9 forces a recalculation, so F9 redraws every simulation trial at once.

draw from a chosen distribution.

The source is `=RAND()`, which returns a uniform draw u on $[0, 1]$. The shaping step is the inverse-CDF method: feed u into the inverse of the target distribution's cumulative function. For the triangular distribution with minimum a , most likely value c , and maximum b — the workhorse of chapter 14, because engineers can estimate those three numbers — the standard inverse-CDF draw is

$$x = \begin{cases} a + \sqrt{u(b-a)(c-a)}, & u < \frac{c-a}{b-a}, \\ b - \sqrt{(1-u)(b-a)(b-c)}, & \text{otherwise,} \end{cases} \quad (21.1)$$

which types directly, with a , b , c , and u standing for the cell references that hold them:

$$\begin{aligned} &=IF(u < (c-a)/(b-a), a + SQRT(u*(b-a)*(c-a)), \\ &\quad b - SQRT((1-u)*(b-a)*(b-c))). \end{aligned}$$

Anchor the three parameters with dollar signs, leave u relative, and the formula fills down into a column of independent draws.

Example 21.17 — Ten trials of the crushing unit

Simulate the Wadi Phosphate crushing unit of Chapter 12, keeping the investment (55,000 SAR), salvage (5,000 SAR), life (5 years), and MARR (10%) fixed, but treating the annual revenue as triangular (36,000; 42,000; 46,000) and the operating cost as triangular (24,000; 26,000; 28,000) — the endpoints are the pessimistic and optimistic scenario values of Example 12.5, and the most likely values are the base case. Run ten trials and summarize.

Solution.

Step 1: the trial row.

Each trial needs two uniform draws, one per uncertain input: `=RAND()` in columns B and C. Column D shapes B into a revenue draw with Equation (21.1), column E shapes C into a cost draw, and column F prices the trial with the base-sheet formula of Table 21.14:

$$F5: =-55000 - PV(0.10, 5, D5-E5, 5000),$$

filled down. Ten rows of the grid, with the uniform draws *fixed at the given*

values below so that the table is reproducible — type these u values in place of `RAND()` and the formulas return exactly these numbers (revenue and cost shown to the nearest SAR):

	A	B	C	D	E	F
4	trial	u_R	u_C	revenue	op. cost	PW
5	1	0.13	0.68	38,793	26,400	-4,916.75
6	2	0.52	0.21	41,586	25,296	9,854.81
7	3	0.87	0.49	43,720	25,980	15,352.22
8	4	0.29	0.90	40,171	27,106	-2,365.89
9	5	0.71	0.07	42,594	24,748	15,754.20
10	6	0.05	0.55	37,732	26,103	-7,810.75
11	7	0.44	0.83	41,138	26,834	2,329.09
12	8	0.95	0.16	44,586	25,131	21,852.15
13	9	0.61	0.38	42,050	25,744	9,920.04
14	10	0.33	0.74	40,450	26,558	765.99

Table 21.18. Ten fixed trials of the crushing unit. In the live workbook, columns B and C hold `=RAND()` and every F9 redraws the whole grid.

Step 2: the summary formulas.

Three formulas turn the PW column into the deliverables of chapter 14:

statistic	formula	displays
P ₁₀	<code>=PERCENTILE(F5:F14, 0.1)</code>	-5,206.15
P ₅₀	<code>=PERCENTILE(F5:F14, 0.5)</code>	6,091.95
P ₉₀	<code>=PERCENTILE(F5:F14, 0.9)</code>	16,363.99
loss probability	<code>=COUNTIF(F5:F14, "<0")/COUNT(F5:F14)</code>	0.30

PERCENTILE interpolates between order statistics, so its values need not equal any single trial. COUNTIF with the criterion "<0" counts the losing trials — three of the ten — and dividing by COUNT turns the count into a relative frequency:

$$\Pr(\text{PW} < 0) \approx 0.30 \text{ on these ten trials.}$$

Step 3: read the ten rows plainly.

Ten trials demonstrate the machinery; they do not estimate anything reliably. The companion workbook runs 1,000 trials of the full model of

chapter 14, which also draws the salvage — uniformly between 3,000 and 7,000 SAR — instead of fixing it. There the summary settles near the values that chapter reports from its seeded 10,000-iteration run, $P10 \approx -5,200$, $P50 \approx +6,600$, $P90 \approx +17,200$ SAR, and a loss probability near 0.24 — and still shifts slightly on every F9, as a sampled estimate must. One reading deserves note even in the small grid: the sample mean, $\text{=AVERAGE}(F5:F14) = 6,073.51$ SAR, sits well below the base-case PW of 8,757.19 SAR. That is not sampling noise. The revenue triangle is asymmetric — its mean, $(36,000 + 42,000 + 46,000)/3 = 41,333$ SAR, is below its most likely value of 42,000 SAR — so the expected PW is only about 6,230 SAR. Plugging in most likely values, as the base case does, is not the same as taking the expected value, which is precisely the lesson chapter 14 draws from this model.

Two practical notes close the section. First, iteration without filling a thousand rows: if the model occupies a single row, a one-variable Data Table (Section 21.5) pointed at any *unused* input cell, with the trial numbers 1, 2, . . . , 1,000 as its column, forces one recalculation per trial and records the volatile PW cell each time — a thousand draws from one formula. Second, freezing the draws: because $\text{RAND}()$ is volatile, every edit anywhere on the sheet redraws the sample. To hold one sample still while you work — or to make a graded result reproducible, as Table 21.18 does — copy the draw columns and paste them back as values (Paste Special → Values), which replaces the formulas with the numbers they last displayed. Keep a live copy on another sheet; the frozen one no longer responds to F9.

Reference. Companion workbook 05_simulation_and_trees.xlsx, the DecisionTree sheet.

21.8 Rolling Back a Tree in a Spreadsheet

The rollback procedure of chapter 13 evaluates a decision tree from its tips to its root: at a chance node, take the expected value of the branches; at a decision node, take the best branch and discard the rest. Each rule is one spreadsheet function. SUMPRODUCT multiplies a range of probabilities by a range of payoffs element by element and sums the products — the expected value at a chance node — and MAX over the alternatives' expected values is the decision node's choice. A tree in a spreadsheet is therefore a small grid of states, probabilities, and payoffs, with one SUMPRODUCT per chance node and one MAX per decision node, and the expected value of perfect information (EVPI) falls out as the difference of two cells.

Example 21.18 — The pilot-plant tree, rolled back in six cells

Mirror the pilot-plant decision of chapter 13. Najd Copper can build a full heap-leach plant now: if demand is strong (probability 0.6) the plant’s PW is +40 million SAR; if demand is weak (probability 0.4), –25 million SAR. Alternatively it can spend 5 million SAR on a pilot plant, treated here as a perfect test that reveals the demand state before the build decision, or it can decline the project. Roll the tree back and price perfect information.

Solution.

Step 1: the state grid.

One row per state of the world. Column C holds the payoff of building; column D holds the *informed* choice in that state — build or walk away, whichever is better — which is a decision node at the second stage, so it is a MAX:

	A	B	C	D
1	state	prob.	build	=MAX(C, 0)
2	strong demand	0.6	40	=MAX(C2, 0)
3	weak demand	0.4	-25	=MAX(C3, 0)
4	EV		=SUMPRODUCT(\$B\$2:\$B\$3, C2:C3)	=SUMPRODUCT(\$B\$2:\$B\$3, D2:D3)

Table 21.19. The pilot-plant tree as a grid (payoffs in million SAR). Row 4 is the chance node, one SUMPRODUCT per column; column D applies the second-stage MAX before the probabilities are applied.

Cell C4 displays 14.0: building now is worth $0.6(40) + 0.4(-25) = 14$ million SAR in expectation. Cell D4 displays 24.0: with the state known in advance, the firm builds only in the strong state, $0.6(40) + 0.4(0) = 24$ million SAR.

Step 2: the root decision.

Three alternatives meet at the root:

cell	formula	displays
B6 build now	=C4	14.0
B7 pilot first, then decide	=D4-5	19.0
B8 decline	0	0.0
B9 root value	=MAX(B6:B8)	19.0

Run the pilot plant: root value 19.0 million SAR .

Step 3: EVPI as the difference of two cells.

Perfect information is worth what it adds to the best uninformed decision:

$$=D4 - \text{MAX}(C4, 0) \rightarrow \boxed{\text{EVPI} = 24.0 - 14.0 = 10.0 \text{ million SAR}}.$$

The EVPI is the ceiling on what *any* test can be worth, exactly as chapter 13 defines it, and it explains the rollback verdict in one line: the pilot costs 5 million SAR for information worth at most 10, so it clears its price even before its imperfections are modeled. Raise the pilot's cost cell past 10 and the MAX at the root flips to building now — one input edit, and the whole tree re-decides itself.

Insight. The spreadsheet habit that keeps a tree honest is the same inputs-first rule as everywhere else: probabilities and payoffs live in their own cells, and every node is a formula over them. The common error is typing the rolled-back value of a subtree as a number. That freezes the node: when a probability is revised, the frozen node keeps its stale verdict, and MAX happily selects a branch that is no longer best. If every node is SUMPRODUCT or MAX, the tree re-rolls itself on every edit, which is the entire point of building it in a spreadsheet.

Reference. Companion workbook 06_levelized_cost.xlsx: the LCOE sheet, the Carbon sheet, and the LCC sheet.

21.9 Levelized Cost and Carbon in a Spreadsheet

The levelized cost of energy of Section 6.7 is one PMT call divided by an annual energy: annualize the installed cost per kilowatt, add the fixed O&M, divide by the kilowatt-hours each kilowatt delivers in a year, and add any cost that is already per kilowatt-hour, such as fuel. chapter 17 extends the same calculator by one row — a carbon cost per kilowatt-hour — and this section builds both. The grid below reproduces the remote-mine comparison of Example 6.7 (MARR 8%), one column per option:

	A	B (solar)	C (diesel)
1	MARR	0.08	
2	installed cost (SAR/kW)	6000	1200
3	fixed O&M (SAR/kW-yr)	60	0
4	fuel (SAR/kWh)	0	0.95
5	capacity factor	0.25	0.60
6	life (years)	25	15
7	energy (kWh/kW-yr)	=8760*B5	=8760*C5
8	annualized capex	=PMT(\$B\$1, B6, -B2)	=PMT(\$B\$1, C6, -C2)
9	LCOE (SAR/kWh)	=(B8+B3)/B7+B4	=(C8+C3)/C7+C4

Table 21.20. The LCOE calculator. Row 8 is the (A/P) annualization as a PMT call; row 9 divides by the energy and adds the per-kWh fuel.

Row 7 displays 2,190 and 5,256 kWh per kW-year; row 8 displays 562.07 and 140.20 SAR per kW-year; and row 9 displays 0.2841 against 0.9767 SAR per kWh, the hand values of Example 6.7 (0.284 and 0.977 from four-decimal factors). Because everything hangs off the MARR cell, the sheet answers in one edit the question Section 6.7 flagged as decisive: at a different discount rate, the capital-heavy solar number moves and the fuel-heavy diesel number barely does.

Example 21.19 — An internal carbon price flips the ranking

Mirror the internal-carbon-price analysis of chapter 17. At the mine of Example 6.7, fuel reaches the site at a subsidized price of 0.20 SAR per kWh delivered, not the 0.95 SAR market price, so the diesel column wins on cash alone. The company prices carbon internally at 90 SAR per tCO₂e, and its gensets emit 0.80 kgCO₂e per delivered kWh (the solar plant: zero). Reprice both options, and find the carbon price at which the ranking flips.

Solution.

Step 1: the cash-only ranking.

Change the diesel fuel cell to 0.20. Row 9 now displays 0.2841 (solar) against 0.2267 (diesel): on subsidized fuel, diesel is cheaper by about 0.057 SAR per kWh.

Step 2: the carbon rows.

Add three rows to the grid, with the carbon price in its own input cell B11:

	A	B (solar)	C (diesel)
11	carbon price (SAR/tCO ₂ e)	90	
12	emissions (kgCO ₂ e/kWh)	0	0.80
13	carbon cost (SAR/kWh)	=\$B\$11*B12/1000	=\$B\$11*C12/1000
14	LCOE with carbon	=B9+B13	=C9+C13

The division by 1,000 converts kilograms to tonnes. Row 13 displays 0 and 0.072 SAR per kWh, and row 14 displays

0.2841 (solar) against 0.2987 (diesel) SAR per kWh :

at the internal price of 90 SAR per tCO₂e, the ranking flips and solar wins.

Step 3: the flip price, algebraically.

The ranking changes where the carbon charge closes the cash gap, so the threshold is a plain formula, not a Goal Seek run:

$$=(B9-C9)/(C12/1000) \rightarrow p^* = 71.72 \text{ SAR per tCO}_2\text{e}.$$

Any internal price above 71.72 SAR per tCO₂e selects solar; any price below it selects diesel. Stating the threshold is more useful than stating one verdict, because the internal carbon price is a policy input that management, not the engineer, will set — the sheet’s job is to show exactly where the decision turns. The Carbon sheet of the companion workbook carries this grid, and its LCC sheet applies the same NPV discipline to a two-pump life-cycle-cost comparison (acquisition, energy, maintenance, and end-of-life rows), where the higher-priced efficient pump wins by about 25,300 SAR over ten years.

Reference. Companion workbook 07_estimation.xlsx: the Scaling sheet, the Escalation sheet, and the Contingency sheet.

21.10 *Estimation Sheets*

The cost-estimation methods of chapter 9 are three multiplications, which makes them the easiest sheets in this chapter and the easiest to get silently wrong. Capacity scaling raises a size ratio to a fractional exponent, escalation multiplies by an index ratio, and contingency multiplies by one plus a percentage read from a class table. The spreadsheet contribution is the same as everywhere else: each factor in its own labeled cell, so the estimate updates when the exponent, the index, or the class changes.

Capacity scaling is one POWER call. With the reference capacity Q_1 and cost C_1 , the new capacity Q_2 , and the exponent x in input cells,

$$=C1*POWER(Q2/Q1, x),$$

which is $C_2 = C_1(Q_2/Q_1)^x$, the six-tenths rule when $x = 0.6$. The caret form $=C1*(Q2/Q1)^x$ is the same computation; keep the exponent in a cell either way, because x is an empirical estimate, not a constant of nature. Escalation is the index ratio $=cost*(index_now/index_then)$, and contingency comes from a lookup: $=VLOOKUP(class, table, column, FALSE)$, with the final FALSE forcing an exact match on the class number so a mistyped class returns an error instead of a silently wrong neighbor.

	D	E	F
1	class	stated accuracy	conting.
2	5	−50% / +100%	40%
3	4	−30% / +50%	25%
4	3	−20% / +30%	15%
5	2	−15% / +20%	10%
6	1	−10% / +15%	5%

Table 21.21. An AACE-style estimate-class table for the VLOOKUP, placed in columns D–F beside the pipeline. The accuracy ranges follow the published class definitions; the contingency column holds representative course values — each organization calibrates its own.

Example 21.20 — A crushing-plant estimate, three multiplications

Wadi Lithium is screening a 350 t/h crushing plant. The best reference is a 2019 vendor quote: 8.0 million SAR for a 200 t/h plant of the same configuration. The plant cost index stood at 118.4 at the quote date and stands at 132.7 today, and the study is at AACE Class 4. Build the estimate sheet.

Solution.

	A	B
1	reference capacity Q_1 (t/h)	200
2	reference cost C_1 (SAR, 2019)	8000000
3	new capacity Q_2 (t/h)	350
4	exponent x	0.6
5	scaled cost	=B2*POWER(B3/B1, B4)
6	index at quote date	118.4
7	index today	132.7
8	escalated cost	=B5*B7/B6
9	AACE class	4
10	contingency	=VL00KUP(B9, D2:F6, 3, FALSE)
11	estimate with contingency	=B8*(1+B10)

Table 21.22. The estimation pipeline: scale, escalate, add contingency. The class table of Table 21.21 sits in D2:F6.

Cell B5 displays 11,192,132: the six-tenths rule prices the $1.75\times$ capacity step at a factor of $1.75^{0.6} = 1.3990$, not 1.75 — the economy of scale of chapter 9. Cell B8 escalates by $132.7/118.4 = 1.1208$ to 12,543,884 SAR in today’s money. Cell B10 returns 25% for Class 4, and B11 displays

15,679,855 SAR $\approx$ 15.7 million SAR .

Two readings. First, scaling and escalation are both multiplications, so their order does not change the result; what matters is that each is applied exactly once. Second, report the estimate with its class attached: “15.7 million SAR, Class 4, $-30\%/ +50\%$ ” is an honest number, while “15,679,855 SAR” alone dresses a screening estimate in seven-significant-figure clothing that chapter 9 warns against.

In practice. The exponent 0.6 is a fitted regularity, not a law. Published exponents for mineral-processing equipment run from about 0.5 to 0.8, and using 0.6 for a plant whose true exponent is 0.7 misprices a $1.75\times$ scale-up by about 6%. That is why the exponent belongs in an input cell: the first sensitivity check on any scaled estimate is to sweep x across its plausible range and watch the estimate move.

21.11 *Good Spreadsheet Habits*

The examples of this chapter all follow the same construction rules. They are not cosmetic; each one removes a class of errors that shows up in real project sheets.

- **Inputs live in their own block**, labeled, one estimate per cell, at the top of the sheet. Everything below is formulas. A reviewer, including you in three

months, can then see at a glance what was assumed.

- **No hard-coded numbers inside formulas.** Write `=B$2*E7`, never `=0.08*E7`. A number buried in a formula is invisible, and it silently survives the day the rate changes to 9%.
- **Label the units.** “SAR per year,” “years,” “fraction per year.” Half the errors in student sheets are a rate typed as 8 where a formula expects 0.08.
- **One change propagates everywhere.** This is the test of a well-built sheet: change the MARR cell and every PW, every profile point, and every AEC row updates together. If updating an estimate requires editing three formulas, the sheet is wrong even when its numbers are right.
- **Check against a known answer.** Every sheet in this chapter reproduced a hand-solved example before it was trusted with anything new. Do the same with your own sheets: rebuild a book example first, then edit the inputs toward your real problem.

Insight. Why the exam still requires factor notation when a spreadsheet is faster. A formula like `=PMT(B2,B3,-B1)` shows *that* you got a number, not *why* it is the right number. The factor notation $A = P(A/P, 8\%, 5)$ names the transformation: which amount moved, in which direction, over how many periods, at what rate. That is the part an engineer must get right, because the spreadsheet will cheerfully compute the wrong model to fifteen digits. The course grades the reasoning; the tool only automates the arithmetic that comes after it.

21.12 Chapter Summary

- Excel’s time-value functions share one argument family (rate, nper, pmt, pv, fv, type) and one **sign convention**: money in positive, money out negative. Enter directions the way you draw a CFD; do not patch signs after the fact.
- FV, PV, PMT, RATE, and NPER each solve the equivalence equation for the unknown they are named after, and one call can replace a two-factor hand solution.

- A **loan schedule** is the payment from PMT plus three filled-down formulas (interest on the opening balance, principal, new balance). The final balance must be exactly zero; IPMT/PPMT/CUMIPMT answer single-row questions without the table.
- NPV discounts its range starting at **period 1**: the pattern is `=A0+NPV(rate, A1:AN)`. IRR is the opposite: its range **includes year 0**, assumes equal spacing, and empty cells are not zeros. A PW profile is one anchored NPV formula against a column of rates; an incremental IRR is IRR of a literal difference column.
- SLN, DDB, and VDB reproduce the book depreciation methods, including the salvage floor (built into DDB) and the switch to straight line (built into VDB). MACRS has no function: type the percentage table and multiply with absolute references. The ATCF table is the Chapter 10 pipeline, one column per line.
- A one-variable **Data Table** automates one-way sensitivity; **Goal Seek** solves one-unknown break-even problems; scenarios are plain rows with their own inputs. All three only work if the sheet was built inputs-first.
- The **AEC(N) table** is two PMT calls per row, one for capital recovery (with the resale value as f_v) and one wrapped around NPV for the operating costs, plus MIN and MATCH to flag N^* .
- Table answers and spreadsheet answers differ in the last digits because the tables round factors to four decimals and Excel does not. Small differences are expected; large ones are sign errors or misplaced years.

Table 21.23 collects the whole mapping in one place; it is meant to be used as a study aid next to the formula sheet.

function / tool	computes	book section
FV	future worth of P and/or A	§2.7, §2.11
PV	present worth of A and/or F	§2.7, §2.11
PMT	equal series: (A/P) and (A/F)	§2.11, §6.3
RATE	unknown interest rate	§2.9, §7.3
NPER	unknown number of periods	§2.9
IPMT, PPMT	one payment's interest / principal split	§2.11
CUMIPMT	accumulated loan interest	§2.11
NPV	PW of a range starting at year 1	§5.4
IRR	i^* of a range starting at year 0	§7.3
IRR of a difference column	incremental rate i_{B-A}^*	§7.9
SLN	straight-line charge	§10.3
DDB	declining balance, salvage floor included	§10.4
VDB	DDB with switch to straight line	§10.4
percentage table $\times$ basis	MACRS schedule	§10.7
Data Table	one-way sensitivity	§12.3
Goal Seek	break-even value of one input	§12.4
scenario rows	scenario analysis	§12.5
PMT twice per row	$AEC(N)$ table	§16.4
MIN, MATCH	flag the economic service life N^*	§16.4
RAND	uniform draw on $[0, 1)$, one per uncertain input	Ch. 14
IF + SQRT	triangular draw from a RAND value	Ch. 14
PERCENTILE	$P_{10} / P_{50} / P_{90}$ of a simulated column	Ch. 14
COUNTIF(..., "<0")/n	probability of loss	Ch. 14
SUMPRODUCT	expected value at a chance node	Ch. 13
MAX	the choice at a decision node; EVPI	Ch. 13
PMT over annual energy	LCOE, with or without carbon	§6.7, Ch. 17
POWER	capacity scaling $(Q_2/Q_1)^x$	Ch. 9
VLOOKUP	contingency from a class table	Ch. 9

Table 21.23. Every method in the book mapped to its spreadsheet counterpart, with the section where the hand method lives.

21.13 Exercises

These exercises are “reproduce and extend” tasks: each one rebuilds a sheet from this chapter against a hand answer you already have, then asks a what-if question the rebuilt sheet answers in one input edit. They follow the chapter's order, from the five time-value functions and loan schedules through NPV and IRR, depreciation, Data Tables and Goal Seek, and the $AEC(N)$ table, and they are worked at a computer, typing every formula rather than copying from the companion workbooks.

The five functions and loan schedules

Exercise 21.1. Rebuild the FV grid of Example 21.1 for the robotics research fund (100,000 SAR at 7%). Then answer two what-if questions with one formula each: (a) the fund balance after 12 years instead of 8; (b) the number of years for the fund to reach 200,000 SAR, checked against the Rule of 72.

Exercise 21.2. Rebuild the amortization schedule of Example 21.6 for the phosphate plant loan of exercise Exercise 2.11: 60,000 SAR at 9%, five equal end-of-year payments. Report (a) the payment, (b) the balance after year 2, (c) the interest portion of the year-3 payment by IPMT and confirm it against the schedule, and (d) the total interest over the loan, compared with the hand value 17,127.74 SAR.

NPV, IRR, and the incremental method

Exercise 21.3. A project's cash flows sit in B3:B8 with the year-0 flow in B3, and the MARR sits in B1. Which formula computes the project's net present worth?

- (A) =NPV(B1, B3:B8)
- (B) =B3 + NPV(B1, B4:B8)
- (C) =NPV(B1, B4:B8) - B3
- (D) =IRR(B3:B8)

Exercise 21.4. Rebuild the crusher comparison of Example 5.8 in a three-column grid (C1, C2, and the increment C2−C1; both 5-year lives, MARR 10%; C1: −80,000, +26,000/yr, salvage 10,000; C2: −110,000, +34,000/yr, salvage 15,000). Report each machine's PW and IRR, and the incremental IRR of C2−C1, and state the choice. Which machine has the higher IRR?

Exercise 21.5. In Example 7.3, the conveyor drive's rate of return was found by hand: bracket between 18% and 20%, then interpolate to 18.78%. Rebuild the PW cell (−75,000 now, +24,400 SAR per year for 5 years, rate in its own input cell) and find the exact i^* with Goal Seek. Compare the Goal Seek answer, the IRR answer, and the hand interpolation, and explain the pattern of the differences.

Depreciation, taxes, and uncertainty

Exercise 21.6. The filling machine of exercise Exercise 10.1 has a cost basis of 90,000 SAR, a salvage value of 10,000 SAR, and an 8-year life. Build a schedule with an SLN column and a DDB column. In which year does DDB's salvage floor bind, and what is the capped charge that year?

Exercise 21.7. Rebuild the bagging-line ATCF sheet of Example 21.12, with the tax rate in its own input cell, and change the rate from 25% to 30%. The other inputs are unchanged: basis 60,000 SAR, straight line over 4 years to 12,000 SAR, revenue 50,000 SAR, cash expenses 20,000 SAR, sale at 15,000 SAR, MARR 10%. Report the new operating ATCF, the new net proceeds of the sale, and the new present worth. Does the verdict change?

Exercise 21.8. Rebuild the crushing-unit base sheet of Table 21.14 and its revenue Data Table (Example 21.13). (a) Between which two rows of the table does the PW change sign? (b) Use Goal Seek to find the exact break-even revenue and confirm the 5.5% cushion of Example 12.4. (c) Why can a Data Table not answer the scenario question of Example 21.15?

Replacement

Exercise 21.9. Extend the ore-transfer-truck AEC sheet of Example 21.16 by one row: the team estimates that holding the truck a fifth year would end with a market value of $S_5 = 21,000$ SAR and cost $OC_5 = 26,000$ SAR in operations. Compute $AEC(5)$, and state whether the economic service life changes. What do MIN and MATCH return now?

Exercise 21.10. A colleague's AEC sheet computes capital recovery as $\text{=PMT}(\text{\$B\$2}, A5, -\text{\$B\$1})$, with no fourth argument, and the sheet recommends keeping every asset as long as possible. Diagnose the error and state the correct formula.

Simulation, trees, levelized cost, and estimation

Exercise 21.11. Rebuild the ten-trial simulation grid of Example 21.17, typing the ten fixed (u_R, u_C) pairs of Table 21.18 in place of $\text{RAND}()$. (a) For trial 1 ($u_R = 0.13$),

state which branch of Equation (21.1) the revenue formula takes and verify the 38,793 SAR draw by hand. (b) chapter 14 also treats the salvage as uncertain, uniform between 3,000 and 7,000 SAR. Write the one-line formula that shapes a uniform draw u_S into a salvage value, add the column to the grid, and recompute trial 1's PW with $u_S = 0.85$. (c) The ten-trial P50 is 6,091.95 SAR; the 1,000-trial workbook settles near 6,600 SAR. Which is the better estimate, and why is the gap not evidence of a construction error?

Exercise 21.12. Rebuild the pilot-plant grid of Example 21.18, then revise two estimates: the weak-demand payoff worsens from -25 to -35 million SAR, and the pilot plant's cost rises to 6 million SAR. (a) Report the new values of the build-now cell, the informed cell, and the root MAX, and state the decision. (b) Report the new EVPI. (c) Using the EVPI, state without further computation the pilot cost at which the root decision flips to building now, and say how the sheet confirms it.

Exercise 21.13. Two one-edit questions on the sheets of this chapter's last two sections. (a) In the carbon grid of Example 21.19, a vendor offers newer gensets that emit 0.65 instead of 0.80 kgCO₂e per delivered kWh, with everything else unchanged (subsidized fuel at 0.20 SAR per kWh). Recompute the flip price and state whether the internal price of 90 SAR per tCO₂e still selects solar. (b) In the estimation sheet of Example 21.20, the team concludes that plants of this configuration scale with exponent 0.7, and the study has matured to AACE Class 3. Report the new estimate with contingency and compare it with the Class-4 figure of Example 21.20.

Exam-style problems

Exercise 21.14. An analyst at *Najd Copper* lays out the evaluation of a pit dewatering system: 70,000 SAR at year 0, net returns of 21,000 SAR in years 1–4, and 30,000 SAR in year 5 (the last net plus a 9,000 SAR salvage), at a MARR of 12%. The sheet reads:

	A	B
1	MARR	0.12
2	year	cash flow
3	0	-70000
4	1	21000
5	2	21000
6	3	21000
7	4	21000
8	5	30000
9	PW?	=NPV(B1, B3:B8)
10	PW?	=B3+NPV(B1, B4:B8)
11	IRR?	=IRR(B3:B8)
12	IRR?	=IRR(B4:B8)

(a) (8 pts) Which of B9 and B10 computes the project's present worth? Predict what each cell displays. (b) (5 pts) State the exact algebraic relation between the two displayed values and explain where it comes from. (c) (6 pts) One of the two IRR cells displays 17.76% and the other displays an error. Which is which, and why? (d) (6 pts) State the accept/reject verdict by the PW rule and by the IRR rule, and explain why the two rules must agree here.

Exercise 21.15. A colleague builds a declining-balance schedule for *Wadi Phosphate's* dryer feed pump: cost 64,000 SAR in B1, salvage 8,000 SAR in B2, life 5 years in B3, and years 1–5 down A6:A10. The charge column holds

$$C6: =DDB(\$B\$1, 0, \$B\$3, A6),$$

filled down, and the check cell holds C12: =SUM(C6:C10). (a) (8 pts) Predict the five charges the sheet displays and the book value it implies at the end of year 5. (b) (4 pts) What does the check cell display, and what must a correct schedule's total be? (c) (5 pts) Name the bug and the rule it violates. (d) (8 pts) State the corrected formula, the corrected year-4 and year-5 charges, and the corrected ending book value.

Assessing outside recommendations

Spreadsheets from outside arrive with their verdicts attached. The two problems below are audits: decide whether to accept the recommendation, name every

flaw, and correct the numbers. The flawed work is quoted exactly as received.

Exercise 21.16. Wadi Phosphate asked an AI assistant to evaluate a filter-press rebuild: 88,000 SAR now; net savings of 27,000 SAR in years 1, 2, and 4; year 3 is a scheduled overhaul year with a net of exactly zero; and 36,000 SAR in year 5 (the last net plus salvage). The MARR is 10%. The assistant returned this sheet and note:

	A	B
1	MARR	0.10
2	0	-88000
3	1	27000
4	2	27000
5	3	
6	4	27000
7	5	36000
9	PW	=NPV(B1, B2:B7)

“Cell B9 displays +3,394.08. The present worth at 10% is positive, so the rebuild clears the MARR and should be approved.”

Accept or reject the sheet? Name each flaw and correct the number.

Exercise 21.17. A consultant compares leasing and buying a mobile light tower for *Najd Copper’s* night operations, at $i = 10\%$ (the rate sits in cell B2). Buying: 38,000 SAR now (B1), resale 14,000 SAR after 4 years (B4), and operating costs of 6,000 SAR per year. Leasing: a full-service contract at 15,800 SAR per year, operating costs included. The consultant’s sheet computes capital recovery as

$$=PMT(\$B\$2, 4, -\$B\$1, -\$B\$4),$$

which displays 15,004.48, and the memo reads:

“Owning costs 15,004 SAR per year in capital recovery plus 6,000 SAR in operations, 21,004 SAR per year in total, against 15,800 SAR per year for the lease. Leasing also leaves the 38,000 SAR purchase price free for other uses, a further advantage. Recommendation: lease.”

Accept or reject the memo? Name each flaw and correct the numbers.